Basic Exercises about *Mathematica*

1. Calculate $\sqrt{5}$ with four decimal places.

```
N[√5]
√5.
```

```
2.23607
```

```
2.23607
```

- We can evaluate a cell by placing the cursor on it and pressing Shift+Enter (or Enter on the numeric key pad).

- If we want to calculate $\sqrt{5}$ and we type just “$\sqrt{5}$”, *Mathematica* gives the exact value of $\sqrt{5}$, which is $\sqrt{5}$.

```
√5
```

```
√5
```

- To get a numerical value of $\sqrt{5}$, we type N[$\sqrt{5}$]. The N[...] command tells *Mathematica* to evaluate the quantity in brackets numerically. N[...,n] displays n digits.

```
N[√5, 10]
```

```
2.236067977
```

- Square brackets [...] are used for enclosing arguments in commands or functions. Parenthesis (...) are used for grouping and braces {...} are used to enclose components of arrays and elements of sets (in general to make a list).

- Palettes allow you to easily insert mathematical notation. For example, to write the cube root of 34, you could click on $\sqrt[\square]{\blacksquare}$ and type 34 and 3 inside the squares (clicking on them) .

- Whenever you give a number with an explicit decimal point, *Mathematica* produces an approximate numerical result.

2. Evaluate the following cell and write the expression providing the third component of the array *v*:

```
v = {2.3478, 4.4449, 5.7902, 7.1126, 9.8855};
```

```
v[[3]]
```

```
5.7902
```

- A semi-colon (;) at the end of a line will suppress the output. *Mathematica* does the computation but does not print it to the screen.

```
v = {2.3478, 4.4449, 5.7902, 7.1126, 9.8855} (*without semi-colon*)
```

```
{2.3478, 4.4449, 5.7902, 7.1126, 9.8855}
```

- In order to make reference to a component of an array, use double square brackets.

3. Solve the equation $t \ln(t) - 3t + 10 = 6$.

```
Solve[t Log[t] - 3 t + 10 == 6, t]
Solve[t Log[t] - 3 t + 10 == 6., t]
N[Solve[t Log[t] - 3 t + 10 == 6, t]]
```

Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. ≫

$$\left\{\left\{t \to -\frac{4}{\text{ProductLog}\left[-\frac{4}{e^3}\right]}\right\}, \left\{t \to -\frac{4}{\text{ProductLog}\left[-1, -\frac{4}{e^3}\right]}\right\}\right\}$$

Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. ≫

```
{{t → 1.56883}, {t → 15.5229}}
```

Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. ≫

```
{{t → 15.5229}, {t → 1.56883}}
```

- The command Solve[...] is used to solve equations or systems of equations. The first letter of a command is always a capital letter.
- We enter ln(t) as Log[t].
- We use == to define equations. The symbol = defines assignations.
- *Mathematica* gives the solution as a set of rules (we will speak later about rules).
- Remember that when any number in an arithmetic expression is given with an explicit decimal point, you get an approximate numerical result for the whole expression.

4. Plot the function $t \ln(t) - 3t + 10$ on the interval [0,30] and verify that it takes the value 6 exactly twice.

```
Plot[t Log[t] - 3 t + 10, {t, 0, 30}]
```



```
Plot[{6, t Log[t] - 3 t + 10}, {t, 0, 30}]
```



5. Add options to the previous plot to force the range to be [0,30]. What is the effect of adding the option AspectRatio->Automatic?

```
Plot[t Log[t] - 3 t + 10, {t, 0, 30}, PlotRange → {0, 30}]
Plot[t Log[t] - 3 t + 10, {t, 0, 30}, PlotRange → {0, 30}, AspectRatio → Automatic]
```





- The operation of many *Mathematica* commands can be influenced by a variety of options of the form “option name->special option setting”.

- AspectRatio is an option for Plot and other graphic functions. If AspectRatio is set to a number it specifies the height to width ratio of the resulting graphic. If the AspectRatio is set to Automatic, *Mathematica* sets the width and height so that objects will not be distorted (we can use it, for example, if we want to draw a circle which looks like a circle).

6. Solve the system of equations:

$$x^2 - 3y^2 = 10$$

$$\frac{x}{y} - \frac{3}{y} = 1$$

$$x + y + \ln(z) = 7$$

```
Solve[{x^2 - 3 y^2 == 10, x / y - 3 / y == 1, x + y + Log[z] == 7}, {x, y, z}]
Solve[{x^2 - 3 y^2 == 10, x / y - 3 / y == 1, x + y + Log[z] == 7.}, {x, y, z}]
```

$$\left\{\left\{x \to \frac{1}{2}\left(9-\sqrt{7}\right),\ y \to \frac{1}{2}\left(3-\sqrt{7}\right),\ z \to e^{1+\sqrt{7}}\right\},\ \left\{x \to \frac{1}{2}\left(9+\sqrt{7}\right),\ y \to \frac{1}{2}\left(3+\sqrt{7}\right),\ z \to e^{1-\sqrt{7}}\right\}\right\}$$

```
{{x → 3.17712, y → 0.177124, z → 38.3115}, {x → 5.82288, y → 2.82288, z → 0.192868}}
```

7. Define the functions $f(x) = \frac{x^3}{x^4+1}$, $g(x, y) = \sqrt{25 - x^2 - y^2}$ and $h(x, y) = (x + 2\,y, \text{xy})$. Calculate $f(5)$, $g(1, 2)$ and $h(f(3), 2)$.

```
f = Function[x, x^3/(x^4 + 1)];
g = Function[{x, y}, Sqrt[25 - x^2 - y^2]];
h = Function[{x, y}, {x + 2 y, x y}];
f[5.]
g[1, 2.]
h[f[3], 2.]
```

```
0.199681
```

```
4.47214
```

```
{4.32927, 0.658537}
```

8. Plot $f(x)$ on the interval $[-5, 5]$.

```
Plot[f[x], {x, -5, 5}]
```



9. Plot in the same graph $f(x)$ and $f'(x)$.

```
Plot[{f[x], f'[x]}, {x, -5, 5}]
```



- To obtain $f'(x)$ simply type f'[x].

10. Solve the equation $f'(x) = 0$. Relate the result with both functions of the previous plot.

```
Solve[f'[x] == 0, x]
Solve[f'[x] == 0., x]
Solve[f'[x] == 0., x, Reals]
```

$\{\{x \to 0\}, \{x \to 0\}, \{x \to -3^{1/4}\}, \{x \to -i\, 3^{1/4}\}, \{x \to i\, 3^{1/4}\}, \{x \to 3^{1/4}\}\}$

Solve::ratnz : Solve was unable to solve the system with inexact coefficients. The answer was obtained by solving a corresponding exact system and numericizing the result. ≫

```
{{x → -1.31607}, {x → 0.}, {x → 0.},
 {x → 0. - 1.31607 i}, {x → 0. + 1.31607 i}, {x → 1.31607}}
```

Solve::ratnz : Solve was unable to solve the system with inexact coefficients. The answer was obtained by solving a corresponding exact system and numericizing the result. ≫

```
{{x → -1.31607}, {x → 0.}, {x → 1.31607}}
```

11. Plot the function $g(x, y)$ of Exercise 7 for $-5 \le x \le 5$ and $-5 \le y \le 5$.

```
Plot3D[g[x, y], {x, -5, 5}, {y, -5, 5}]
```



12. Generate a list of the squares of the numbers from 1 to 10.

```
Table[n^2, {n, 1, 10}]
```

```
{1, 4, 9, 16, 25, 36, 49, 64, 81, 100}
```

- The Table command can be used to generate a list of numbers using a predefined mathematical expression. It defines an array of objects satisfying a given condition.

- The loop variable does not have to be an integer. A list of evenly spaced numbers in the interval between 0 and 1 can be generated by:

```
Table[x, {x, 0, 1, 0.1}]
```

```
{0., 0.1, 0.2, 0.3, 0.4, 0.5, 0.6, 0.7, 0.8, 0.9, 1.}
```

Hence the iterator {x, 0, 1, 0.1} indicates that the lower limit on the variable x is 0, the upper limit is 1, and the interval between successive values of x is 0.1.

13. Generate a list of pairs $\{n, \sqrt{n}\}$ for n between 1 and 20. Plot the pairs you have obtained.

```
Table[{n, N[√n]}, {n, 1, 20}]
ListPlot[%]
```

```
{{1, 1.}, {2, 1.41421}, {3, 1.73205}, {4, 2.}, {5, 2.23607},
 {6, 2.44949}, {7, 2.64575}, {8, 2.82843}, {9, 3.}, {10, 3.16228},
 {11, 3.31662}, {12, 3.4641}, {13, 3.60555}, {14, 3.74166}, {15, 3.87298},
 {16, 4.}, {17, 4.12311}, {18, 4.24264}, {19, 4.3589}, {20, 4.47214}}
```



- In this example, the expression used to construct the list ($\{n, \sqrt{n}\}$) is itself a list.

- We can plot a graph of these points with the command ListPlot. The modifier Joined -> True tells Mathematica to connect the points with lines.

```
ListPlot[Table[{n, √n}, {n, 1, 20}], Joined → True]
```



- The symbol % is used to refer to the last output given by *Mathematica*.

- ListPlot[{y_1, y_2, ..., y_n}] plots points corresponding to a list of values, assumed to correspond to x coordinates 1, 2, … . That is, it is equivalent to ListPlot[{{1,y_1},{2,y_2}, ..., {n,y_n}}]. We can set the range of the variable x by using DataRange->{a,b}.

```
ListPlot[{2, 2.5, 3, 3.5, 3, 2.5, 2}]
ListPlot[{2, 2.5, 3, 3.5, 3, 2.5, 2}, DataRange -> {0, 6}]
```





14. Plot the list of pairs (sin(n), sin(2 n)) for n between 0 and 100. Compare with the curve (sin(t), sin(2 t)) for t varying in the interval [0,7].

```
ListPlot[Table[{Sin[i], Sin[2 i]}, {i, 0, 100}], AspectRatio → Automatic]
ParametricPlot[{Sin[i], Sin[2 i]}, {i, 0, 7}, AspectRatio → Automatic]
```





- Notice that if a command consists of several words, the first letter of each word comprising the command is a capital letter (ListPlot[...], ParametricPlot[...], AspectRatio->, DataRange->, etc.)

15. Solve the differential equation:

$$\frac{dS}{dt} = 0.03148\, S\,, \quad (\text{satisfying} \quad S(0) = 46\,612\,)$$

```
DSolve[{S'[t] == 0.03148 S[t], S[0] == 46612}, S, t]
```

$\{\{S \to \text{Function}[\{t\}, 46612.\ e^{0.03148\,t}]\}\}$

```
F = S /. %[[1]]
```

$\text{Function}[\{t\}, 46612.\ e^{0.03148\,t}]$

- In order to solve differential equations, we use the command DSolve.
- Remember that you need to use == to define equations. If you type = instead, *Mathematica* will produce an error warning message and even if you type == to fix the mistake, *Mathematica* will show another error message. You can sort it out by evaluating Remove[S]. In general, when you get funny error messages, you can try with Clear["Global`*"] or Remove["Global`*"] to clear all variables. Sometimes it is also useful to close *Mathematica* and open it again.
- $S \to \text{Function}[\{t\}, 46612.\ e^{0.03148\,t}]$ is a "rule" that can be applied to S by using the command /. and its effect is the replacement of S (the left hand side) by $\text{Function}[\{t\}, 46612.\ e^{0.03148\,t}]$ (the right hand side).
- The solution is given as an array of solutions. In this case, we have a single solution, which in turn is an array comprising a single function expressed as a rule. We refer to this solution by using %[[1]].
- Now F is the solution function and we can evaluate it, plot it, etc:

```
F[2]
```

49641.

```
Plot[F[t], {t, 0, 4}]
```



- Once you get the solution, you can also use the option "get solution" that *Mathematica* gives you in the suggestion bar:

```
DSolve[{S'[t] == 0.03148 S[t], S[0] == 46612}, S, t]
```

$\{\{S \to \text{Function}[\{t\}, 46612.\ e^{0.03148\,t}]\}\}$

$\{\{S \to \text{Function}[\{t\}, 46612.\ e^{0.03148\,t}]\}\}[[1, 1, 2]]$
```
(*This is written by Mathematica when you click on "get solution"*)
```

$\text{Function}[\{t\}, 46612.\ e^{0.03148\,t}]$

```
F = %;
F
```

```
Function[{t}, 46612. e^(0.03148 t)]
```

$\text{Function}\left[\{t\},\ 46\,612.\ e^{0.03148\, t}\right]$

16. Solve the logistic differential equation for an initial value x_0 and plot the solution for $r = 1.2$ and $x_0 = 0.3$.

$$\frac{d\,x}{d\,t} = r\,x\,(1 - x)$$

```
DSolve[{x'[t] == r x[t] (1 - x[t]), x[0] == x0}, x, t ]
x = x /. %[[1]];
x[t]
x0 = 0.3; r = 1.2;
Plot[x[t], {t, 0, 10}]
```

Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. ≫

$$\left\{\left\{x \to \text{Function}\left[\{t\},\ \frac{e^{r\,t}\,x0}{1 - x0 + e^{r\,t}\,x0}\right]\right\}\right\}$$

$$\frac{e^{r\,t}\,x0}{1 - x0 + e^{r\,t}\,x0}$$